

```

// FreeFEM code for Phantom B

real R2=0.00852; // outer radius [m]
real R1=0.00708; // inner radius [m]
real Rd=0.00778; // notch outer radius [m]
real delta=17.3*pi/180.; // half of notch angle

border S1(t=2*pi,0){x=R1*cos(t); y=R1*sin(t); label=1;} //inner surface
border S2a(t=0,3*pi/2-delta){x=R2*cos(t); y=R2*sin(t); label=2;} // S2a-e outer surface
border S2b(t=R2,Rd){x=t*cos(3*pi/2-delta);y=t*sin(3*pi/2-delta); label=2;}
border S2c(t=3*pi/2-delta,3*pi/2+delta){x=Rd*cos(t); y=Rd*sin(t); label=2;}
border S2d(t=Rd,R2){x=t*cos(3*pi/2+delta);y=t*sin(3*pi/2+delta); label=2;}
border S2e(t=3*pi/2+delta,2*pi){x=R2*cos(t); y=R2*sin(t); label=2;}

int n=360;
int nc=40;
mesh Th=buildmesh( S1(n)+S2a(3*n/4-nc/2)+S2b(5)+S2c(nc)+S2d(7)+S2e(n/4-nc/2));

plot(Th,ps="tube.ps"); // tube shape before deformation
fespace Ph(Th,P1);

real Ec = 0.275*10^6; // Young's modulus [Pa]
real nuc = 0.495; // Poisson's Ratio
real muc = Ec/(2*(1+nuc)); // Lame's second parameter
real lambdac = Ec*nuc/((1+nuc)*(1-2*nuc)); // Lame's first parameter
Ph E = Ec;
Ph nu = nuc;
Ph mu = muc;
Ph lambda = lambdac;

real p1=0.101*10^6+98.0665*18; // pressure inside [Pa] overpressure is 18cmH2O
real p2=0.101*10^6; // pressure outside, atmospheric pressure [Pa]

real sqrt2=sqrt(2.);
macro epsilon(u1,u2) [dx(u1),dy(u2),(dy(u1)+dx(u2))/sqrt2] // EOM

```

```

macro div(u,v) ( dx(u)+dy(v) ) // EOM

fespace Vh(Th,[P1,P1]);
Vh [uu,vv], [w,s];

solve elas([uu,vv],[w,s])=
int2d(Th)(lambda* div(w,s)* div(uu,vv) +2.*mu*(epsilon(w,s)* epsilon(uu,vv)))
+ int1d(Th,1)( p1 * (w*N.x + s*N.y) )
+ int1d(Th,2)( p2 * (w*N.x + s*N.y) )
+ on(3,uu=0,vv=0);

plot([uu,vv],wait=1);
plot([uu,vv],wait=1,bb=[[-0.5,1],[1,-0.5]]);
mesh th1 = movemesh(Th, [x+uu, y+vv]);
plot(th1,wait=5); // tube after deformation

savemesh(Th,"Tube deform.msh"); // save tube before deformation
savemesh(th1,"Tube deform2.msh"); // save tube after deformation

```